

# HTML::Display - HTML zum lokalen Browser senden

Dank HTML::Display von Max Maischein (corion) lassen sich einzelne HTML-Seiten ohne Web-Server ganz einfach von Perl aus zum Browser schicken und in diesem anzeigen.

Extrem praktisch, wenn man z.B. einen dynamischen generierten Report als HTML-Datei speichern und diesen gleich im lokalen Browser anzeigen lassen möchte. Fast schon eine Fernsteuerung für Browser.

HTML::Display funktioniert auf den gängigen Betriebssystemen mit den meisten Browsern, wie z.B. Internet Explorer oder Firefox. Bisher habe ich es allerdings nur mit Windows XP SP3 und Ubuntu 8.04 ausprobiert.

### Beispiel: lokale HTML-Datei zum Browser des Users senden

```
#!/usr/bin/perl
use strict;
use warnings;
use HTML::Display; # importiert display()

my $file = "index.html";

display( file => $file );
```

```
<html>
<head>
<title>
HTML-Display rocks
</title>
</head>
<body>
<h1>HTML-Display rocks</h1>
<p>
Lorem ipsum dolor sit amet, consetetur sadipscing elitr,
sed diam nonumy eirmod tempor invidunt ut labore et
dolore magna aliquyam erat, sed diam voluptua.
</p>
<p>
Duis autem vel eum iriure dolor in hendrerit in vulputate velit
esse molestie consequat, vel illum dolore eu feugiat nulla
facilisis at vero eros et accumsan et iusto odio dignissim qui
blandit praesent luptatum zzril delenit augue dui dolore te
feugait nulla facilisi.
</p>
</body>
</html>
```

Die dazugehörige HTML-Datei `index.html` im selben Verzeichnis (siehe Listing 1).

### Beispiel: Webseite zum Browser des Users senden

Dazu wird zunächst die Website mit LWP::Simple geholt und dann an HTML::Display weitergereicht.

```
#!/usr/bin/perl
use strict;
use warnings;

use HTML::Display;
use LWP::Simple;

my $location = "http://perl-howto.de/";

my $html = get $location;

display( html => $html,
        location => $location );
```

Der Parameter `location` setzt die Navigationsleiste des Browsers, während `html` den Quelltext der anzuzeigenden Seite erwartet.



**Beispiel: Datenbankreport auf die Schnelle im Browser anzeigen**

Aus einer (sehr einfachen) Adress-Datenbank

```
$ sqlite3 Adressen.db
sqlite> select customer_id,vorname,
        nachname from customer;
1|Hans|Meier
2|Susanne|Schmidt
3|Otto|Muster
```

werden alle Datensätze ermittelt, anschließend mit HTML-Tags versehen und im lokalen Browser angezeigt (siehe Listing 2).

Geht schnell und benötigt keinen Web-Server!

# Thomas Fahle

```
#!/usr/bin/perl
use strict;
use warnings;
use DBI;
use Encode;          # importiert decode_utf8
use HTML::Display;   # importiert display()
use CGI;

my $table = retrieve();

my $html = create_html( $table );

display( html => $html ); # legt automatisch eine temp-Datei an

sub create_html {
    my $table = shift @_;
    my $cgi = new CGI;
    my $content;
    $content = $cgi->start_html( 'Database Report' );
    $content .= $cgi->start_table( { -border => 1 } );

    foreach my $row (@$table) {
        $content .= $cgi->Tr( { -align => 'left', -valign => 'top' },
                             $cgi->td( $row )
                             );
    }

    $content .= $cgi->end_table();
    $content .= $cgi->end_html();

    $content = decode_utf8( $content );
    return $content;
}

sub retrieve {

    my $dbh = DBI->connect( "dbi:SQLite:dbname=Adressen.db", "", "",
                           { RaiseError => 1 } );

    my $sql = qq~select customer_id,vorname,nachname from customer~;

    my $sth = $dbh->prepare( $sql );
    $sth->execute();

    my $tbl_ary_ref = $sth->fetchall_arrayref;

    $sth->finish();
    undef $sth;
    $dbh->disconnect();

    return wantarray ? @$tbl_ary_ref : $tbl_ary_ref;
}
```

**Listing 2**